import QtQuick 1.1
import com.nokia.symbian 1.1
import com.nokia.extras 1.1

PageStackWindow {
    id: window
    initialPage: MainPage {tools: toolBarLayout}
    showStatusBar: true
    showToolBar: true

    ToolBarLayout {
        id: toolBarLayout
        ToolButton {
            flat: true
            iconSource: "toolbar-back"
            onClicked: window.pageStack.depth <= 1 ? Qt.quit() : window.pageStack.pop()
        }
        ToolButton {
            flat: true
            iconSource: "toolbar-menu"
            onClicked: menu.open();
        }
    }

    Menu {
        id: menu

        MenuLayout {
            MenuItem {
                text: qsTr("_adicionar")
                onClicked: {
                    window.pageStack.push(Qt.resolvedUrl("NovoPage.qml"));
                }
            }
            MenuItem {
                text: qsTr("_excluir")
                enabled: trackModel.existemItensSelecionados
                onClicked: {
                    trackModel.RemoverTracksSelecionados();
                }
            }
            MenuItem {
                text: qsTr("_atualizar")
                enabled: trackModel.existemItensSelecionados
                onClicked: {
                    trackModel.AtualizarTracksSelecionados();
                }
            }
            MenuItem {
                text: qsTr("_ajuda")
                onClicked: {
                    window.pageStack.push(Qt.resolvedUrl("AjudaPage.qml"));
                }
            }
            MenuItem {
                text: qsTr("_sobre")
                onClicked: {
                    window.pageStack.push(Qt.resolvedUrl("SobrePage.qml"));
                }
            }
        }
    }

    Connections {
        target: internet

        onPesquisaIniciada: {
            dlgProgresso.open();
        }

        onPesquisaConcluida: {
            dlgProgresso.close();
            dlgConsultaOk.open();
        }
    }    

    DlgProgresso {
        id: dlgProgresso

        onCancelar: internet.CancelarConsulta();
    }

    QueryDialog {
        id: dlgConsultaOk
        titleText: qsTr("_pesquisa_concluida_dlg_titulo")
        message: qsTr("_pesquisa_concluida_dlg_mensagem")
        acceptButtonText: qsTr("_pesquisa_concluida_dlg_btn_ok")
        onAccepted: {
            dlgConsultaOk.close();
        }
    }
}
